

## Paper 218-27

## Fun With Timelines: Doing More With SAS/GRAPH® Proc GPLOT

Barbara B. Okerson, Ph.D., Virginia Health Quality Center, Glen Allen, VA

### ABSTRACT

SAS® PROC GPLOT is designed as a tool for the plotting and visualization of data values. The built-in flexibility of the procedure, together with the ease of combining output with the SAS Annotate facility also makes it a good tool for creating a variety of timelines. Timelines are an essential part of process completion and process evaluation whether in the public, private, academic or government sector. Included in this paper are four examples of different types of timelines as well as the code used to create these examples. All examples were developed with version 8.1 of SAS under the Windows 98 operating system.

### INTRODUCTION

The Virginia Health Quality Center (VHQC), winner of the U.S. Senate Productivity and Quality Award, is a health care quality improvement organization that assists healthcare providers in making successful and meaningful changes in the way care is delivered and in improving outcomes of that care, especially for the Medicare community of Virginia. Services provided by VHQC include assisting healthcare organizations with performance improvement techniques, reviewing health plan denials, statistical consulting and data analysis services, health education, credentials verification, clinical and claim database engineering, health care utilization and quality reviews, and coding/DRG validations. VHQC currently collaborates with all acute care hospitals in Virginia to improve the quality of health care.

Like many organizations, VHQC is continuously faced with meeting shorter deadlines with higher quality output. While this challenge can only be met by careful planning and scheduling of time, this process can be greatly assisted by the planning, plotting and measuring of the time involved for the processes required for completion of these tasks. Timelines become a valuable process tool for all aspects of healthcare quality improvement.

As illustrated in the examples below, timelines can be used to list events by time, plot milestones and deadlines, identify conflicting priorities, measure actual time for process completion, compare methods for timeliness of results, and display all of these in a user-friendly and functional format.

The examples that follow do not use actual data for confidentiality reasons, but are presented as an illustration of the use of timelines for process decision-support. No specialty software is needed, as SAS provides all the tools needed for the creation of a large variety of timelines. Although additional timelines can be created with SAS software, especially with SAS/QC, SAS/OR, Base SAS Proc TIMEPLOT and other SAS/GRAPH procedures, this paper presents only those created with the SAS/GRAPH GPLOT procedure.

### EXAMPLE ONE

In this example, a simple timeline is created showing system processes running across an 8 hour time period. By considering a timeline of scheduled processes, new production runs can be timed to optimize resources. Here at VHQC, large data sets are received on a regular basis and merged with an existing

encounters database. This process is extremely resource-intensive; other resource-intensive processes need to avoid this scheduled time period when possible. In this example, resource-intensive processes are displayed in red with other processes displayed in green.

As illustrated in the code that follows, Proc GPLOT is used to plot the beginning and end points of the processes on a two-dimensional axis. A SAS Annotate Data Set is created to draw and color the lengths of the process times on the GPLOT.

SAS Program:

```

/* Set Graphics Options*/
goptions ftext=zapf targetdevice=jpeg
device=win;

/* Create Data Set*/
data process;
input Process $18. @20 begin time5.
    @26 end time6. linetype;
format begin end time5.;
cards;
System Check          08:00 09:00  1
Production Run One    09:30 13:00  1
Virus Scan            11:59 13:30  1
Production Run Two    13:00 16:00  1
Backup                17:00 18:00  1
;
run;

/*Create Annotate Data Set to Draw Lines*/
data anno;
length function color $8;
retain xsys ysys '2' size 10
    color 'green'; set process;
line=linetype;
if process="Backup" then color='red';
function='move'; x=begin;
    yc=Process; output;
function='draw'; x=end; yc=process;
    output;run;
title "Information Systems Automated
    Process Schedule";


/* Set Symbols and Axes*/
symbol1 i=none;
axis1 order=("Backup" "Production Run Two"
    "Virus Scan" "Production Run One"
    "System Check") label = none;
axis2 order=(28800 to 64800 by 3600)
    minor=none label=("Scheduled Time");

/*Plot the Timeline*/
proc gplot annotate=anno;
plot process*end/vaxis=axis1 haxis=axis2;
format end time5.;
run;

```

Example One Output:

This output and the others that follow in this paper were created using the SAS Export as Image selection from the graphics output window.




All examples provided in this paper can be customized. To customize the above to emphasize the end points of each process timeline, the following statements can be changed in the creation of the SAS Annotate Data Set.

```
data anno;
  length function color $8;
  retain xsys ysys '2' size 2 color 'green';
  set process;
  line=linetype;
  if process="Backup" then color='red';
  function='move'; x=begin; yc=process; output;
  function='draw'; x=end; yc=process; output;
  function='symbol'; text='dot'; output;
run;
```

In the output that follows, the top boundary line of the plot was also removed by adding the noframe option to the Proc Gplot Plot statement.

Customized timeline:



## EXAMPLE TWO

The second example is similar to the above, but allows for the processes to be intermittent and adds reference lines to the plot. This timeline provides a planning tool for an analytical department by identifying when data is collected, and therefore providing a means of estimating when data will be received for analysis. At VHQC, this type of timeline can be used to track the collection of baseline, interim, and remeasurement hospital data abstractions, and to block time for analysis and report generation. It also can be used to provide a reference of when data is being collected for each clinical indicator.

Because many of the reports that are generated at VHQC provide quarterly information, reference lines delineating calendar quarters were added to this timeline plot example. As in the previous example, the output could be modified or customized to support multiple colors, to include symbols, and to include descriptive text as part of the graphic.

SAS Program:

```
/*Set Graphics Options*/
goptions ftext=zapf htext=.9 htitle=2;


/*Create Data Set*/
data prevent;
input #1 Process $22. linetype
#2 @1 (Begin End End2 Begin3 End3
      Begin4 End4) (monyy5.);
cards;
Agency One Pretests          1
feb00 may00 jul00 aug00 jan01 may01 jul01 aug01
Agency Two Pretests          1
apr00 sep00 apr01 sep01
Agency One Posttests         1
mar00 jun00 sep00 oct00 mar01 jun01 sep01 oct01
Agency Two Posttests         1
jun00 dec00 jun01 dec01
Agency One Followups         3
oct00 jan02
Agency Two Followups         3
jul00 dec01
;
run;

/*Create Annotate Data Set*/
data anno;
length function color $8;
retain xsys ysys '2' size 5 color 'red';
set prevent;
line=linetype;
function='move'; x=Begin; yc=Process; output;
function='draw'; x=End; yc=Process; output;
function='move'; x=Begin2; yc=Process; output;
function='draw'; x=End2; yc=Process; output;
function='move'; x=Begin3; yc=Process; output;
function='draw'; x=End3; yc=Process; output;
function='move'; x=Begin4; yc=Process; output;
function='draw'; x=End4; yc=Process; output;
run;

/*Set Symbols and Axes*/
axis1 order=(14641 to 15401 by 90)
label=('Begin - End') minor=none;
axis2 label=none order=("Agency Two Followups"
  "Agency One Followups" "Agency Two Posttests"
  "Agency One Posttests" "Agency Two Pretests"
  "Agency One Pretests");
symbol1 i=none;
title c=blue'Prevention Evaluation Time Line';

/*Create the timeline*/
proc gplot data=prevent;
plot Process*End Process*End2 Process*End3
      Process*End4/ name="time"
      overlay anno=anno haxis=axis1 vaxis=axis2
      href=14670 14762 14854 14945 15035 15127
      15219 15310 lhref=4 chref=gray;
format end end2 end3 end4 monyy5.;
run;
```

Example Two Output:



### EXAMPLE THREE

In this example, a historical style timeline is created that tracks events planning for a training session. The X-Y axis is created with SAS/GRAPH Proc GPLOT, with the needle interpolation option used to mark the event points on the X-axis. The points are labeled with a SAS Annotate Data Set.

#### SAS Program:

```

/*Create Data Set*/
data timeline;
input @1 label $32. @34 date mmdyy10.;
space=2; format date mmdyy10.;
cards;
First Planning Meeting          09/27/2001
Reserve Meeting Places         09/30/2001
Contract With Caterers        10/03/2001
Mail Invitations to Agencies  10/10/2001
Agency RSVP Deadline         10/19/2001
Finalize Training Materials    10/23/2001
Order Needed Supplies         10/25/2001
Reserve Laptop and Projector   11/01/2001
Finalize Materials            11/06/2001
Work Party: Assemble Notebooks 11/10/2001
Training Conference Site 1     11/13/2001
Training Conference Site 2     11/15/2001
;
run;

/*Set Graphics Options*/
goptions border targetdevice=jpeg ftext=swiss
border;

/*Create Annotate Data Set*/
data anno;
retain function 'label' color 'red'
position '3' hsys '3'
ysys '2' xsys '2' size 3 style 'swissi';
set timeline;
text=label;x=date+.5; y=space;angle=70;
run;

/*Set Symbols and Axes*/
title2 c=blue h=2
"Prevention Agency Training Schedule 2001";
axis1 order=(1 to 100 by 1) major=none
minor=none value=none style=0 label=none;
axis2 c=blue width=10 major=(height=1.5)
label=none order=(15245 to 15295 by 5)
value=(height=1);
symbol c=blue i=needle;

```


```

/*Plot the Timeline*/
proc gplot data=timeline annotate=anno;
plot space*date/vaxis=axis1 haxis=axis2;
format date mmdyy5.
;run;

```

#### Example Three Output:

The resulting timeline follows. This type of timeline is often referred to as a historical timeline and can be used to identify dates relating to any process or event. To further enhance the usefulness of this type of timeline, different types of events can be identified by color, and a hierarchy of event importance can be created through font size.



### EXAMPLE FOUR

This example shows comparison times for several processes that began at the same time. This type of graph would normally be created with SAS/GRAPH Proc GCHART rather than Proc GPLOT, but can easily be created with GPLOT by using a SAS Annotate Data Set to draw and connect the plotted lines. This type of timeline can be used, not only to monitor internal processes, but also to monitor the effectiveness over time of different methods of quality improvement.

#### SAS Program:

```

/* Set Graphics Options*/
goptions ftext=zapf targetdevice=jpeg
device=win;

/* Create Data Set*/
data Runtime;
input @12 runtime $5. @24 begin time5.
@30 end time5. linetype;
format begin end time5.;
cards;
Runtime ID 86478          08:00 09:00 1
Runtime ID 75723         08:00 13:00 1
Runtime ID 48574         08:00 14:30 1
Runtime ID 79684         08:00 10:30 1
Runtime ID 41241         08:00 12:30 1
Runtime ID 41029         08:00 09:30 1
Runtime ID 25079         08:00 11:59 1
;
run;

```

```

/*Create Annotate Data Set to Draw Lines*/
data anno;
  length function color $8;
  retain xsys ysys '2' size 32 color 'green';
  set Runtime;
  line=linetype;
function='move'; x=begin; yc=Runtime; output;
function='draw'; x=end; yc=Runtime; output;
run;
title1
  "Competing Information Systems Processes";
title2 "Testing for Efficiency";


/* Set Symbols and Axes*/
symbol1 i=none;
axis1 label=none;
axis2 order=(28800 to 57600 by 3600) minor=none
label=("Process Run Time");

/*Plot the Timeline*/
proc gplot annotate=anno;
plot Runtime*end/vaxis=axis1 haxis=axis2
  href=36000 43200 50400 lhref=4;
format end time5.;
run;

```

#### Example Four Output:

In this example, horizontal reference lines are used to mark time boundaries. Enhancements could include a colored reference point of acceptability, with results past that line changing color. Results could also be sorted by elapsed time.



## CONCLUSION

By using the built-in flexibility of SAS/GRAPH, as well as the capabilities of the SAS Annotate facility, a variety of timelines can be created to measure and ultimately enhance productivity. The examples presented serve as a starting point for a continuing adaptation of the SAS system for time and process measurement.

## REFERENCES

<http://www.vhqc.org>.

Okerson, Barbara B. 1994. "Using the SAS System as a Tool for Historical Research," *Proceedings of the Nineteenth Annual SAS Users Group International Conference*. SAS Institute Inc. Cary, NC. pp. 717-720.

SAS Institute Inc. 1999. *SAS/GRAPH Software: Reference, Version 8, Volumes 1 and 2*.

SAS Institute Inc. 1990. *SAS/GRAPH Software: Reference, Version 6, First Edition, Volumes 1 and 2*.

## ACKNOWLEDGEMENTS

SAS and SAS/GRAPH are registered trademarks of SAS Institute Inc. of Cary, North Carolina in the USA and other countries.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The analyses upon which this publication is based were performed under contract no. 500-99-VA-03, titled "Utilization and Quality Control Peer Review Organization for the Commonwealth of Virginia, sponsored by the Centers for Medicare & Medicaid Services (CMS), Department of Health and Human Services (DHHS). The content of this publication does not necessarily reflect the view or policies of the DHHS, nor does mention of trade names, commercial products, or organizations imply endorsement by the U.S. Government. The Virginia Health Quality Center (VHQC) assumes full responsibility for the accuracy and completeness of the ideas presented. This publication is a direct result of the Health Care Quality Improvement Program initiated by CMS, which has encouraged identification of quality improvement projects derived from analysis of patterns of care and therefore required no special funding on the part of VHQC. Ideas and contributions to the VHQC concerning experience in engaging with the issues presented are welcomed.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. For more information contact:

Barbara B. Okerson, Ph.D.  
 Virginia Health Quality Center  
 4510 Cox Road, Suite 400  
 Glen Allen, VA 23060  
 Phone: 804.289.5320  
 e-mail: [vapro.bokerson@sdps.org](mailto:vapro.bokerson@sdps.org)



VIRGINIA HEALTH QUALITY CENTER